
Introduction to BPM
    

The next several sections provide an introduction to to Business Process Management. 

  
    
  
  What is a Business Process?


Companies large and small have business processes that they regularly execute. Let's just call them "processes" for brevity's sake. Insurance companies have processes to process a claim. Banks have processes to approve loans. The government have processes to award contracts. 
Business processes have the following charateristics:
	A business process is comprised of some number of individual tasks. 
	Tasks are performed in an order prescribed by network of task connections.
	Processes can branch to multiple alternative sets of tasks depending on the state of the process.
	Processes can branch to multiple parallel paths as well.
	There are multiple kinds of tasks. Some tasks are performed by users. Some tasks are performed by calling services. There are many other kinds of tasks as well.
	Processes store data to be used by tasks to perform their functions.
	Processes allow tasks to add data for use by downstream tasks.


  
    
  
  What is Business Process Management (BPM)?


Business Process Management (BPM) is the organized and directed activity of a company to control, monitor and improve the quality of their business processes. They do this with the aid of a BPM platform made available by dozens of commercial vendors.
BPM platforms provide the following functions:
	Provide developers with the ability of define and store process models.
	Provide developers with the ability have certain tasks completed by making service calls.
	Provide users and administrators with the ability to start process instances.
	Provide users with the ability to assign user tasks to specified users.
	Prvide users with the ability to assign user tasks to members of a specified user group.

The goals of Business Process Management are:
	Better Process Reliablibility
	Faster Process Execution
	Reduced Training Requirements
	Incremental Process Improvement
	Provide Process Metrics
	Cost Savings


  
    
  
  When is BPM Applicable


This is difficult to answer exactly, but BPM may be applicable when:
	Processes are performed on a regular basis.
	Successful process execution is crucial to the business.
	Current processes are error prone.
	The speed of process completion is very important.
	Processes are well defined.
	Multiple business groups and/or people participate in process execution.




  

    
Introduction to Mozart
    


  
    
  
  Process Models


The primary concept of Mozart is the Process Model. A process model is an executable representation of a business process. Process models are implemented using a collection of functions that together comprise a Domain Specific Language (DSL) used for the purpose of creating BPM Applications.
If you are a developer working for a bank, you might be asked to develop a process model for handling a customer loan application. A grossly simplified process model for this might look like:
  defprocess "process loan application" do
    user_task("record and check loan application",  group: "customer service")
    user_task("determine loan acceptance",          group: "underwriting")
    user_task("inform customer of result",          group: "customer service")
  end

  
    
  
  Process Model Tasks


As indicated in the example above, a process model is composed of a collection of tasks. Each task represents some work that needs to be completed. To complete a task means to perform the work required of the task.
Mozart provides around a dozen different types of tasks. Tasks are defined using functions in the Mozart DSL.
The task types currently supported are:
	Task Type	Description
	user_task	Performed by a user
	service_task	Performed by calling a service.
	subprocess_task	Performed by calling a subprocess.
	timer_task	Waits for expiration of a timer.
	receive_task	Waits for a subscribed PubSub event.
	send_task	Sends a PubSub event.
	rule_task	Performed by evaluating a set of rules represented in a table.
	case_task	Selects one of many process paths.
	parallel_task	Initiates two or more parallel process paths.
	prototype_task	Has no behavior. Used for prototyping and subbing.
	repeat_task	Repeats a set of tasks while condition is true.
	reroute_task	Reroutes a process off the typical (happy) execution path
	conditional_task	Implements a conditional task along the typical execution path


  
    
  
  BPM Process Module


A BPM Process Module is an Elixir module containing a set of related process models, BPM applications, BPM related type definitions as well as any other normal function definitions. There is an example process application in the Mozart GitHub repository.
Process applications must use Mozart.BpmProcess to gain access to the Mozart DSL:
defmodule MyProcessApplication do
  use Mozart.BpmProcess

  defun send_loan_response(data) do
    # Code to send customer and email loan resonse
  end

  defprocess "process loan application" do
    user_task("record and check loan application", group: "customer service")
    user_task("determine loan acceptance", group: "underwriting")
    service_task("inform customer of result", function: &MyProcessApplication.send_loan_response/1)
  end

  # multiple process model and function definitions

end
Declaring use Mozart.BpmProcess injects the MyBpmApplication.load/0 function into your application. Calling this function will load all of your process models into the Mozart.ProcessService model repository. Use it now to load your process models:
MyProcessApplication.load()

  
    
  
  BPM Applications


Each Elixir BPM module can define zero or more BPM Applications. Here is an example
def_bpm_application("Home Loan", data: "Customer Name,Income,Debt")
Each application specifies a user level name, the name of the top level process definition and a set of required 
input parameters.
The purpose of a BPM application is to aid in intergration with external applications, e.g. GUI applications.

  
    
  
  Parameter type specifications


Each Elixir BPM process module can define any number of parameter type definitions. Here are examples
of the currently supported types:
def_number_type("number param", min: 0, max: 5)
def_choice_type("choice param", choices: "foo, bar")
def_multi_choice_type("multi choice param", choices: "foo,bar,foobar")
def_confirm_type("confirm param")
Type definitions aid in integration with external applications, e.g. GUIs.

  
    
  
  Process Engine


We have said that a Mozart process model is executable. What does that mean? After loading the process model as shown above, commands are invoked to start and begin executing the process model. Here is a corresponding code snippet:
data = %{loan_amount_requested: 300_000}

{:ok, ppid, uid, _business_key} = ProcessEngine.start_process("process loan application", data)

ProcessEngine.execute(ppid)
First, we define some data that the process model will use. Then a process engine is spawned by calling ProcessEngine.start_process/2. The first parameter specifies the top level process model to be executed. As you will see later on, a hierarchy of process instances can be spawned to complete the top level process model. The second argument is data used to initiate process execution. 
The first argument returned is the Elixir process identifier of the Elixir GenServer. The secord value returned is a unique identifier of the process execution. This will be useful beyond the lifetime of the Elixir process. Finally, the third value returned is another unique identifier corresponding to the hierarchy of processes spawned to complete the top level process model.
After, the process engine has been spawned, process execution begins by calling ProcessEngine.execute/1 function, passing it the Elixir PID contained in the return value bound to the variable ppid.

  
    
  
  Process Service


We have already seen that the GenServer Mozart.ProcessService provides a repository for process models. It provides numerous additional functions documented in hexdocs.

  
    
  
  Processes Model Execution


Earlier we said that the job of the process engine is to drive a process model execution to completion. Let's discuss what that means.

  
    
  
  Completing Tasks


When process execution is triggered by calling ProcessEngine.execute/1, the first task specified in the process model will be opened for completion.
 defprocess "process loan application" do
    user_task("record and check loan application", group: "customer service")
    user_task("determine loan acceptance", group: "underwriting")
    service_task("inform customer of result", function: &MyProcessApplication.send_loan_response/1)
  end
Hence, in the process model above, the user task named record and check loan application will be opened for completion upon the initiation of process execution.
Completion of this initial task will typically trigger the opening of additional tasks. Every task that completes will potentially trigger the opening of additional tasks. Eventually, a process model will have no open tasks. This signals that execution of the process model is complete and the corresponding process engine will exit.
So, completing the execution of a process model equates to repeatedly opening and completing tasks until the process as a whole is complete.

  
    
  
  Accumulating Data


Almost every task that is opened will need some data to complete. Further, when a task completes, it will introduce some new data for downstream tasks to use. Thus, to a great extent, completing a processes equates to generating a store of accumulated information.
All of the data generated by task completion is stored in the data property of Mozart.ProcessEngine.

  
    
  
  Producing Side Effects


To be useful, almost all process model executions will produce some side effects. Take the process of processing a bank loan application for example. What are the possible side effects?
	Sending a notice that the loan is approved or declined.
	Depositing money in the customers account.
	Setting up the loan in the banks backend system.


  
    
  
  Conclusion


Hopefully, the mechanics of Business Process Managment is now much clearer. In the newxt section, we will use Mozart to execute business processes.



  

    
First Process Execution
    

Now we will put together what we've lerned so far to execute a process model. 

  
    
  
  To Follow Along..


To follow along create a new Elxir mix project and then add Mozart as a dependency in your mix.exs project file. Your dependency should look like this:
  defp deps do
    [
      {:mozart, "~> 0.4"}
    ]
  end

  
    
  
  Create BPM Application


First you need to create a BPM application module. In it you will specify your process models and any needed supported functions. Add the following file to your project.
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  # Your process models and supporting functions will go here.

end

  
    
  
  Construct the Process Model


First, let us construct a simple process model that will have a single service task. The service task will simply return the sum of two input properties. Update your BPM application module as follows:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  def sum(data) do
    %{sum: data.x + data.y}
  end

  defprocess "add x and y process" do
    service_task("add x and y task", function: &MyBpmApplication.sum/1, inputs: "x,y")
  end

end

  
    
  
  Load the Process Model into repository


Now let's load this process model into the process model repository, which is implemented by Mozart.ProcessService. First, start an iex session with iex -S mix and paste a couple of aliases into your iex session:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS

Now we are ready to load our process model into the repository. Copy the following into your iex session:
MyBpmApplication.load()

You should see:
iex [15:06 :: 3] > MyBpmApplication.load()
{:ok, ["add x and y process"]}
The function MyBpmApplication.load/0 was created for you automatically when you inserted use Mozart.BpmProcess into your module definition.

  
    
  
  Start a Process Engine and Execute the Process Model


Now let us start a process engine, initializing it with our process model and some data. Then we call the function that will invoke process execution. Copy the following into your iex session:
{:ok, ppid, uid, _key} = PE.start_process("add x and y process", %{x: 1, y: 1})

You should now see something like:
10:04:26.242 [info] Start process instance [add x and y process][a57d93d4-fc97-4a3c-89f6-989c088c96a7]
{:ok, #PID<0.457.0>, "a57d93d4-fc97-4a3c-89f6-989c088c96a7", "d41dc645-2417-4778-bb18-fccf2e6e44f6"}
We started a process instance using Mozart.ProcessEngin.start_process/2. We passed it the name of the process model to be executed and some initial data required for process execution. The function call returned a tuple with three values: 
	The pid of the resulting process engine.
	The uid of the process instance. Every process execution will be given a unique identifier. You will see how this is used going forward.
	Finally, a business_key is returned. Process models are hierarchial, meaning they can be composed of multiple, nested subprocesses. The top level business process and all of its subprocesses will share the same business_key. Here, were aren't going to be using the process key, but we will in subsequent examples.

After creating the process engine instance, we then call Mozart.ProcessEngine.execute/1 which causes the process engine to start completing any tasks that are ready to be completed. Let's do that now:
PE.execute(ppid)

After calling the ProcessEngine.execute function, you should see something like:
iex [08:50 :: 23] > PE.execute(ppid)
:ok
10:02:40.292 [info] New service task instance [add x and y task][ddd36158-7837-4c5a-b0be-66a02f304d27]
10:02:40.292 [info] Complete service task [add x and y task[ddd36158-7837-4c5a-b0be-66a02f304d27]
10:02:40.292 [info] Process complete [add x and y process][a1d4525d-96cb-4da1-9a82-d20bab4afa89]
The logs show that:
	A new service task was opened.
	The service task completed.
	And finally the process engine completed executing the process model.


  
    
  
  Verify the Results


Let's verify the results of our process model execution by calling ProcessService.get_completed_process_data/1. Copy the following into your iex session:
PS.get_completed_process_data(uid)

and you should see:
iex [08:50 :: 26] > PS.get_completed_process_data(uid)
%{sum: 2, y: 1, x: 1}
We see that a sum property was added to process data whose value is the sume of the properties x and y..



  

    
Mozart Task Types
    

In this guide we will provide examples of all of the Mozart BPM DSL functions.
If you want to follow along, create an Elixir mix project that has Mozart as a dependency.
  defp deps do
    [
      {:mozart, "~> 0.4"}
    ]
  end
Now, create an Elixir module named my_bpm_application.ex with the following contents:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  # All of the examples will go here

end


  
    
  
  Service Task


A Service Task performs its work by calling an Elixir function. This function could perform a computation, call an external JSON service, retrieve data from a database, etc.
A service task has two unique arguments: function and inputs.
	The function field specifies the function that the service task should apply for the purpose of returning output data into the process state. The function can be either a captured function or an atom corresponding to a locally
defined function of arity 1.
	The inputs field is used to select which process data fields are passed to the task's function. If no value is supplied for this field, the entire process data is passed.


  
    
  
  Service Task example


In your MyBpmApplication module, enter the following code:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  ## Service Task Example

  def sum(data) do
    %{sum: data.x + data.y}
  end

  defprocess "add x and y process" do
    service_task("add x and y task", function: &MyBpmApplication.sum/1, inputs: "x,y")
    # You could also specify:
    # service_task("add x and y task", function: :sum, inputs: "x,y")
  end

end

We've defined a process model that has a single service task. The name of the task is "add x and y task". It will apply the function MyBpmApplication.sum/1 to the arguments x and y.
Now open an iex session on your project:
$ iex -S mix

Now, paste in a couple of alias's:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS

Now, you need to load your process model into the system. Paste in the following:
MyBpmApplication.load()

which gives the result:
iex [16:41 :: 1] > MyBpmApplication.load()
{:ok,
 ["add x and y process"]}
MyBpmApplication.load/0 loads your BPM process application models into the model repository. You didn't define this function in your module. It was defined for you when you added use Mozart.BpmProcess to your module.
Now you are ready to start a ProcessEngine that will run your process model:
{:ok, ppid, uid, business_key} = PE.start_process("add x and y process", %{x: 1, y: 1})

You should then see:
08:33:47.895 [info] Start process instance [add x and y process][299f3213-94ab-418d-bee3-18da04f2c38d]
{:ok, #PID<0.302.0>, "299f3213-94ab-418d-bee3-18da04f2c38d",
 "4d6e7fba-b303-4c9b-8cf5-a291168bd73d"}
Your process engine is now ready to execute your process model. Run the following function:
iex [08:21 :: 5] > PE.execute(ppid)
:ok
08:37:46.418 [info] New service task instance [add x and y task][211308eb-fc1f-45f9-b758-077b12e920db]
08:37:46.418 [info] Complete service task [add x and y task[211308eb-fc1f-45f9-b758-077b12e920db]
08:37:46.418 [info] Process complete [add x and y process][299f3213-94ab-418d-bee3-18da04f2c38d]
As the log entires indicate:
	A new service task was created.
	The service task completed it work (by applying the function you specified).
	And finally the process completed.

To verify, that your process did what it was supposed to do, execute the following:
iex [08:21 :: 6] > PS.get_completed_process_data(uid)
%{sum: 2, y: 1, x: 1}

  
    
  
  User Task


A User Task works by having a user complete some required task. This often takes the form of a user using a GUI to examine a subset of process data and then supplying additional data. 
Important Note: Users interact with a BPM platfrom such as Mozart by way of end user GUI application. The user application will allow the user to find tasks appropriate to his role. Once the user accepts responsibility for a task, the application will then provide an input form appropriate for accomplishing the given task. For now, we will use Mozart functions for finding and completing tasks.
A user task has two unique arguments: group and :inputs.
	The group field specifies the user group in the form of a comma separated list that are elibigle to complete the task. 
	The inputs field is used to select which process data fields are passed to the user that will compleete the task. If no value is supplied for this field, the entire process data is passed.


  
    
  
  User Task example


First, let's add a process model containing a user task;
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  ## Previous Examples Omitted

  ## User Task Example

  defprocess "one user task process" do
    user_task("user gives sum of x and y", group: "admin", inputs: "x,y")
  end

end

Now we drop back into iex:
iex -S mix
Now we paste in code to define alias's, load our process models, start a process engine and then execute our process model.
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("one user task process", %{x: 1, y: 1})
PE.execute(ppid)

and we should see:
Interactive Elixir (1.16.2) - press Ctrl+C to exit (type h() ENTER for help)
iex [10:39 :: 1] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [10:39 :: 2] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [10:39 :: 3] > MyBpmApplication.load()
{:ok, ["add x and y process", "one user task process"]}
iex [10:39 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("one user task process", %{x: 1, y: 1})

10:39:49.816 [info] Start process instance [one user task process][714ff531-762f-4677-a357-dda88d7fb1c1]
{:ok, #PID<0.285.0>, "714ff531-762f-4677-a357-dda88d7fb1c1",
 "75a5124a-a51c-446f-80e7-d253dd3b5bc9"}
iex [10:39 :: 5] > PE.execute(ppid)
:ok

10:39:49.824 [info] New user task instance [user gives sum of x and y][3fc1cdf8-4049-40db-9172-9a7f58a85cb8]
The last log message tells us that a new user task was opened, but we don't see logs indicating that the task completed or that the process completed. That is because a user task waits to be completed by a user.
To complete the user task we need the task uid, which was printed in the last log statement:
10:39:49.824 [info] New user task instance [user gives sum of x and y][3fc1cdf8-4049-40db-9172-9a7f58a85cb8]
It is the value in last pair of square brackets: "3fc1cdf8-4049-40db-9172-9a7f58a85cb8". 
The final thing we need to complete the task is the data that we want to merge into the process data. We want to merge in the sum of the inputs x and y. A GUI application for completing user tasks would have access to this data and would make it available to the user completing the task. So, let's assume that we know the value of our inputs and we use them to complete our task.
Let's do that now by calling ProcessService.complete_user_task/2. 
PS.complete_user_task("3fc1cdf8-4049-40db-9172-9a7f58a85cb8", %{sum: 3})

and we should see:
iex [13:05 :: 8] > PS.complete_user_task("7f3d7009-ab44-4955-a501-677e0a8a353b", %{sum: 3})
:ok
13:16:21.326 [info] Complete user task [user gives sum of x and y][7f3d7009-ab44-4955-a501-677e0a8a353b]
13:16:21.327 [info] Process complete [one user task process][fca5f283-0992-4921-ba16-f9d902f9e403]

  
    
  
  Subprocess Task Example


The Subprocess Task is completed by calling a subprocess. When subprocess completes, the corresponding subprocess task also completes.
Update your MyBpmApplication module with the following code:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  ## Previous Code Here

  ## Subprocess Task Example

  def add_one_to_value(data) do
    Map.put(data, :value, data.value + 1)
  end

  defprocess "two service tasks" do
    service_task("service task 2", function: &MyBpmApplication.add_one_to_value/1, inputs: "value")
    service_task("service task 3", function: &MyBpmApplication.add_one_to_value/1, inputs: "value")
  end

  defprocess "subprocess task process" do
    service_task("service task 1", function: &MyBpmApplication.add_one_to_value/1, inputs: "value")
    subprocess_task("subprocess task", process: "two service tasks")
  end

end

Now, we will open an iex session and to the following:
	Paste in our alias'.
	Load our process models.
	Start a process engine with the process named "subprocess task process".
	Execute the process.

Open your iex session and paste in:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("subprocess task process", %{value: 0})
PE.execute(ppid)

And you should see something like:
iex [13:50 :: 1] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [13:50 :: 2] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [13:50 :: 3] > MyBpmApplication.load()
{:ok,
 ["add x and y process", "one user task process", "two service tasks",
  "subprocess task process"]}
iex [13:50 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("subprocess task process", %{value: 0})

13:50:13.910 [info] Start process instance [subprocess task process][903561d2-2f27-4014-9bf9-4334a0d93466]
{:ok, #PID<0.288.0>, "903561d2-2f27-4014-9bf9-4334a0d93466",
 "d16c2892-7419-4381-b060-e4d899d24ee6"}
iex [13:50 :: 5] > PE.execute(ppid)
:ok
13:50:13.912 [info] New service task instance [service task 1][ee364170-99ad-4ff9-8eda-e4c598e3d7da]
13:50:13.912 [info] Complete service task [service task 1[ee364170-99ad-4ff9-8eda-e4c598e3d7da]
13:50:13.913 [info] New sub_process task instance [subprocess task][251b46d7-bf0e-4818-bf28-45a628b85087]
13:50:13.913 [info] Start process instance [two service tasks][cd9cb03f-f50c-48f2-ba51-9ed4d31c18fa]
13:50:13.913 [info] New service task instance [service task 2][1f09acb8-b1ef-465e-9fe3-8f9a754f55e2]
13:50:13.913 [info] Complete service task [service task 2[1f09acb8-b1ef-465e-9fe3-8f9a754f55e2]
13:50:13.913 [info] New service task instance [service task 3][31bc8c0b-cd65-4e80-9afe-74fa546c7412]
13:50:13.913 [info] Complete service task [service task 3[31bc8c0b-cd65-4e80-9afe-74fa546c7412]
13:50:13.913 [info] Complete subprocess task [subprocess task][251b46d7-bf0e-4818-bf28-45a628b85087]
13:50:13.913 [info] Process complete [two service tasks][cd9cb03f-f50c-48f2-ba51-9ed4d31c18fa]
13:50:13.913 [info] Process complete [subprocess task process][903561d2-2f27-4014-9bf9-4334a0d93466]
If we examine the logs, we should see that what we expect to happen did happen.
When top level process finishes, we would expect the value parameter would have a value of three. Let's check if that is the case:
iex [13:50 :: 6] > PS.get_completed_process_data(uid)
%{value: 3}

  
    
  
  Case Task Example


The Case Task provides a way to specify alternate execution paths depending on the current state of process execution.
Add the following content to the MyBpmApplication module:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  ## Previous content here

  ## Case Task Example

  def add_two_to_value(data) do
    Map.put(data, :value, data.value + 2)
  end

  def subtract_two_from_value(data) do
    Map.put(data, :value, data.value - 2)
  end

  def x_less_than_y(data) do
    data.x < data.y
  end

  def x_greater_or_equal_y(data) do
    data.x >= data.y
  end

  defprocess "two case process" do
    case_task "yes or no" do
      case_i &MyBpmApplication.x_less_than_y/1 do
        service_task("1", function: &MyBpmApplication.subtract_two_from_value/1, inputs: "value")
        service_task("2", function: &MyBpmApplication.subtract_two_from_value/1, inputs: "value")
      end
      case_i :x_greater_or_equal_y do
        service_task("3", function: &MyBpmApplication.add_two_to_value/1, inputs: "value")
        service_task("4", function: &MyBpmApplication.add_two_to_value/1, inputs: "value")
      end
    end
  end

end
The first argument to case_task is a task name and the second argument is a block of cases.
The first argument to case_i is either a captured function of arity 1 or an atom corresponding to a locally defined function of arity 1. The second argument is a block of tasks.
The process named "two case process" performs this logic:
	If x is less than y, subtract 2 from the value parameter twice.
	if x is greater or equal to y, add 2 to the value parameter twice.

Let's try it out. Open an iex session, and paste in the following:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("two case process", %{x: 1, y: 2, value: 10})
PE.execute(ppid)

and you should see:
iex [14:33 :: 1] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [14:33 :: 2] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [14:33 :: 3] > MyBpmApplication.load()
{:ok,
 ["add x and y process", "one user task process", "two service tasks",
  "subprocess task process", "two case process"]}
iex [14:33 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("two case process", %{x: 1, y: 2, value: 10})

14:33:41.084 [info] Start process instance [two case process][4fb6889e-250c-407e-a332-f904f491b39d]
{:ok, #PID<0.285.0>, "4fb6889e-250c-407e-a332-f904f491b39d",
 "f5a76516-1509-4e01-99dd-949dba2c06c4"}
iex [14:33 :: 5] > PE.execute(ppid)
:ok
14:33:41.088 [info] New case task instance [yes or no][4a7fd5c1-eaa5-4cfc-9edf-73bd28eaa9cb]
14:33:41.088 [info] Complete case task [yes or no][4a7fd5c1-eaa5-4cfc-9edf-73bd28eaa9cb]
14:33:41.088 [info] New service task instance [1][6482abb3-c56f-43d7-870f-83789b77135b]
14:33:41.088 [info] Complete service task [1[6482abb3-c56f-43d7-870f-83789b77135b]
14:33:41.089 [info] New service task instance [2][7ee68e1e-9823-4e92-9613-e34f1c279c3a]
14:33:41.089 [info] Complete service task [2[7ee68e1e-9823-4e92-9613-e34f1c279c3a]
14:33:41.089 [info] Process complete [two case process][4fb6889e-250c-407e-a332-f904f491b39d]
Based on our input data, since x was less than y, 2 should have been subtracted from 10 twice, leaving the value parameter with a value of 6. Let's see if that matches our result:
iex [13:50 :: 6] > PS.get_completed_process_data(uid)
%{value: 6, y: 2, x: 1}

  
    
  
  Send and Receive Task Example


For this example, we will use both the Send and Receive tasks since they are typically used together, that is, a receive task receives a messge from a send task.
Update the MyBpmApplication module with the following content:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  ## Previous Content Here

  ## Send and Receive Example

  def receive_loan_income(msg) do
    case msg do
      {:barrower_income, income} -> %{barrower_income: income}
      _ -> nil
    end
  end

  defprocess "receive barrower income process" do
    receive_task("receive barrower income", selector: &MyBpmApplication.receive_loan_income/1)
  end

  defprocess "send barrower income process" do
    send_task("send barrower income", message: {:barrower_income, 100_000})
  end

end

In this example we will start the "receive barrower income process" first. The "receive barrower income" task will wait until it receives a message of the form . When it receives this message it will merge %{barrower_income: income} into the processes data.
The purpose of the "send barrower income process" task is to send the message that the receive task is waiting for.
Open an iex session, and paste in the following:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("receive barrower income process", %{})
PE.execute(ppid)

and you should see:
iex [17:23 :: 1] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [17:23 :: 2] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [17:23 :: 3] > MyBpmApplication.load()
{:ok,
 ["add x and y process", "one user task process", "two service tasks",
  "subprocess task process", "two case process",
  "receive barrower income process", "send barrower income process"]}
iex [17:23 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("receive barrower income process", %{})

17:23:16.604 [info] Start process instance [receive barrower income process][d9534434-acc0-43a3-83ac-2f6db76cfb3d]
{:ok, #PID<0.296.0>, "d9534434-acc0-43a3-83ac-2f6db76cfb3d",
 "82d4ac32-e192-4a4b-a5e9-29f02c7ee46a"}
iex [17:23 :: 5] > PE.execute(ppid)
:ok

17:23:16.608 [info] New receive task instance [receive barrower income][8286148c-a756-4db9-b21b-86acb3e8d17a]
At this point, we've started the process with the receive task and we see that the expected receive task has been opened. Now we need to run the process with the send task so the waiting receive task can complete. To do that, copy the following into your iex session. 
{:ok, s_ppid, s_uid, s_business_key} = PE.start_process("send barrower income process", %{})
PE.execute(s_ppid)

Notice that when we call PE.start_process/2, we choose different varaible names so that our previous variable values won't be overwirtten.
You should see this result:
iex [17:23 :: 6] > {:ok, ppid, uid, business_key} = PE.start_process("send barrower income process", %{})

17:28:56.665 [info] Start process instance [send barrower income process][d58a5b10-1a12-48f5-9426-9ab90701e933]
{:ok, #PID<0.299.0>, "d58a5b10-1a12-48f5-9426-9ab90701e933",
 "7e9657d1-21ba-40e4-8c42-00f913c62926"}
iex [17:23 :: 7] > PE.execute(ppid)
:ok
17:28:56.666 [info] New send task instance [send barrower income][273a9f29-783c-4306-97c4-eb9e12735a80]
17:28:56.667 [info] Complete send event task [send barrower income[273a9f29-783c-4306-97c4-eb9e12735a80]
17:28:56.667 [info] Complete receive event task [receive barrower income]
17:28:56.667 [info] Process complete [send barrower income process][d58a5b10-1a12-48f5-9426-9ab90701e933]
17:28:56.667 [info] Process complete [receive barrower income process][d9534434-acc0-43a3-83ac-2f6db76cfb3d]
Now we can check whether the expected data was merged into the process data of the first process:
iex [17:38 :: 8] > PS.get_completed_process_data(uid)
%{barrower_income: 100000}

  
    
  
  Parallel and Prototype Task Example


A Parallel Task provides the ability to start two or more parallel execution paths. A Prototype Task has no behavior. A process engine will complete it automatically after it is opened. It is primarily used for stubbing tasks that will be replaced later with one of the other task types.
Copy the following content into MyBpmApplication module:
defmodule MyBpmApplication do
  @moduledoc false
  use Mozart.BpmProcess

  ## Previous Content Here

  ## Parallel and Prototype Task Example

  defprocess "two parallel routes process" do
    parallel_task "a parallel task" do
      route do
        prototype_task("prototype task 1")
        prototype_task("prototype task 2")
      end
      route do
        prototype_task("prototype task 3")
        prototype_task("prototype task 4")
      end
    end
  end

end

Now open an iex session and copy is the following:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("two parallel routes process", %{})
PE.execute(ppid)

You should see:
iex [18:29 :: 1] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [18:29 :: 2] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [18:29 :: 3] > MyBpmApplication.load()
{:ok,
 ["add x and y process", "one user task process", "two service tasks",
  "subprocess task process", "two case process",
  "receive barrower income process", "send barrower income process",
  "two parallel routes process"]}
iex [18:29 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("two parallel routes process", %{})

18:29:26.105 [info] Start process instance [two parallel routes process][08dba7e3-820e-4ea4-985e-5184b8041c80]
{:ok, #PID<0.323.0>, "08dba7e3-820e-4ea4-985e-5184b8041c80",
 "f4753b53-68ad-404b-ae0c-27bc9737b032"}
iex [18:29 :: 5] > PE.execute(ppid)
:ok
18:29:26.108 [info] New parallel task instance [a parallel task][f7aa1870-e01a-46fb-8e0e-9a4353001241]
18:29:26.108 [info] Complete parallel task [a parallel task]
18:29:26.108 [info] New prototype task instance [prototype task 1][609a01d8-c7bd-4539-ad4e-720951ab56a3]
18:29:26.109 [info] New prototype task instance [prototype task 3][16cc3d11-d8cb-4111-9b53-439a4cb10b94]
18:29:26.109 [info] Complete prototype task [prototype task 3]
18:29:26.109 [info] New prototype task instance [prototype task 4][e03c4cb7-e16d-45d4-98c1-4380bd57d240]
18:29:26.109 [info] Complete prototype task [prototype task 1]
18:29:26.109 [info] New prototype task instance [prototype task 2][48402cb3-9274-4da8-9b50-13e0a8b4b77e]
18:29:26.109 [info] Complete prototype task [prototype task 2]
18:29:26.109 [info] Complete prototype task [prototype task 4]
18:29:26.109 [info] Process complete [two parallel routes process][08dba7e3-820e-4ea4-985e-5184b8041c80]
Examime the log messages. You should be able to see that two execution paths were proceeding in parallel.

  
    
  
  Repeat Task Example


The Repeat Task provides the ability to repeat a set of tasks as long a specified condition holds true.
Copy the following new code into the MyBpmApplication module:
defmodule MyBpmApplication do
  @moduledoc false
  use Mozart.BpmProcess

  ## Previous content here

  def continue(data) do
    data.continue
  end

  defprocess "repeat task process" do
    repeat_task "repeat task", &ME.continue/1 do
      prototype_task("prototype task 1")
      prototype_task("prototype task 2")
      user_task("user task", group: "admin")
    end
    prototype_task("last prototype task")
  end

end

Open an iex session and paste in the following:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("repeat task process", %{continue: true})
PE.execute(ppid)

and you should see:
iex [09:10 :: 4] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [09:10 :: 5] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [09:10 :: 6] > MyBpmApplication.load()
{:ok,
 ["add x and y process", "one user task process", "two service tasks",
  "subprocess task process", "two case process",
  "receive barrower income process", "send barrower income process",
  "two parallel routes process", "repeat task process"]}
iex [09:10 :: 7] > {:ok, ppid, uid, business_key} = PE.start_process("repeat task process", %{continue: true})

09:11:31.919 [info] Start process instance [repeat task process][e3963cb8-410d-4039-ad81-eb6730808783]
{:ok, #PID<0.300.0>, "e3963cb8-410d-4039-ad81-eb6730808783",
 "ba5bded6-fbd9-4690-9643-453ab31457fa"}
iex [09:10 :: 8] > PE.execute(ppid)
:ok
09:11:31.923 [info] New repeat task instance [repeat task][b600cf3d-8f84-4d16-8378-96f0bb152440]
09:11:31.923 [info] New prototype task instance [prototype task 1][c6f6cfca-7097-4931-bea4-61404160e5d1]
09:11:31.923 [info] Complete prototype task [prototype task 1]
09:11:31.923 [info] New prototype task instance [prototype task 2][4c526ace-eb2b-4d7d-b918-83f1deda831f]
09:11:31.923 [info] Complete prototype task [prototype task 2]
09:11:31.924 [info] New user task instance [user task][2557b2e6-d6fa-4172-b25b-a809633e5217]
Now, let's complete the user task specifying continue to be true*, i.e. %{continue: true}. This should cause the repeat tasks to be executed one more time:
PS.complete_user_task("2557b2e6-d6fa-4172-b25b-a809633e5217", %{continue: true})

The output verifies this is the case:
iex [09:10 :: 9] > PS.complete_user_task("2557b2e6-d6fa-4172-b25b-a809633e5217", %{continue: true})
:ok
09:17:20.481 [info] New prototype task instance [prototype task 1][4af087bb-865b-41b5-8d6b-55e6c9511d2b]
09:17:20.481 [info] Complete user task [user task][2557b2e6-d6fa-4172-b25b-a809633e5217]
09:17:20.481 [info] Complete prototype task [prototype task 1]
09:17:20.481 [info] New prototype task instance [prototype task 2][965fa775-6280-43d0-ab50-8a966ed69dac]
09:17:20.481 [info] Complete prototype task [prototype task 2]
09:17:20.481 [info] New user task instance [user task][2e855696-e11d-42e7-89a9-330c2a87b11f]
Now, let's complete the new user task with %{continue: false}. This should cause the repeat to complete, the final prototype_task to complete and then the process to complete:
PS.complete_user_task("2e855696-e11d-42e7-89a9-330c2a87b11f", %{continue: false})

And we see the following, which verifies the expected behavior:
iex [09:10 :: 10] > PS.complete_user_task("2e855696-e11d-42e7-89a9-330c2a87b11f", %{continue: false})
:ok
09:23:09.731 [info] Complete user task [user task][2e855696-e11d-42e7-89a9-330c2a87b11f]
09:23:09.732 [info] Complete repeat task [repeat task]
09:23:09.732 [info] New prototype task instance [last prototype task][5b6c63db-46bb-459b-a79f-d0c53a1662ca]
09:23:09.732 [info] Complete prototype task [last prototype task]
09:23:09.732 [info] Process complete [repeat task process][e3963cb8-410d-4039-ad81-eb6730808783]

  
    
  
  Rule Task Example


A Rule Task completes by evaluating a set of rules in a rule table and returning a derived value.
Copy the following new code into the MyBpmApplication module:
defmodule MyBpmApplication do
  @moduledoc false
  use Mozart.BpmProcess

  ## Previous content here

  rule_table = """
  F     income      || status
  1     > 50000     || approved
  2     <= 49999    || declined
  """

  defprocess "single rule task process" do
    rule_task("loan decision", inputs: "income", rule_table: rule_table)
  end

Our table assumes there is a data property named income. If the value of income is greater than 50000, the rule will return a property named status with a value of approved. If income is less than 50000, the value of status will be declined.
[Note: Mozart uses the Tablex library for rule tasks. Tablex is very powerful. Complete documentation is available in hexdocs.]
Open an iex session on your project and paste in:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("single rule task process", %{income: 60_000})
PE.execute(ppid)

and you should see:
iex [16:31 :: 3] > MyBpmApplication.load()
{:ok, .... deleted list of process names}
iex [16:31 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("single rule task process", %{income: 60_000})
16:31:29.857 [info] Start process instance [single rule task process][371b788e-33e6-4d55-9e0c-9e238f552b84]
{:ok, #PID<0.299.0>, "371b788e-33e6-4d55-9e0c-9e238f552b84",
 "83e7bb9a-a732-44b2-bdc3-ae4bbf6c1790"}
iex [16:31 :: 5] > PE.execute(ppid)
:ok
16:31:29.860 [info] New rule task instance [loan decision][ab9ef596-0be3-44de-a66a-6fb41eaf0ab6]
16:31:29.860 [info] Complete run task [loan decision[ab9ef596-0be3-44de-a66a-6fb41eaf0ab6]
16:31:29.863 [info] Process complete [single rule task process][371b788e-33e6-4d55-9e0c-9e238f552b84]
Now let's look at the completed process data to see the property returned from the evaluation of the table:
iex [16:31 :: 6] > PS.get_completed_process_data(uid)
%{status: "approved", income: 60000}
The value of status is "approved", as expected.



  

    
Mozart Event Types
    

Currently, there is just one event type has been implemented (see below). This document will be updated as addtional event types are implemented.

  
    
  
  Task Exit Event


A Task Exit Event allows a process execution to exit an open task in response to an external event (a Phoenix.PubSub event). When the task exits in this way, process execution will follow an alternate execution path as specified the the task exit event.
The task exit event is implemented by the defevent/3 DSL function which takes the following arguments:
	the name of the event
	process: the name of the process that the event will act upon
	exit_task: the name of the task to be exited
	selector: a function that matches on the target event
	do: one or more tasks to be executed when the target task is exited.

  Here is an example:
  defevent "exit loan decision 1",
    process: "exit a user task 1",
    exit_task: "user task 1",
    selector: &BpmAppWithEvent.event_selector/1 do
      prototype_task("event 1 prototype task 1")
      prototype_task("event 1 prototype task 2")
  end

  
    
  
  Task Exit Event Example


In your MyBpmApplication module, enter the following code:
defmodule MyBpmApplication do
  use Mozart.BpmProcess

  ## Previous Content Here

  ## Task Exit Event Example

  def exit_subprocess_task_event_selector(message) do
    case message do
      :exit_subprocess_task -> true
      _ -> nil
    end
  end

  defprocess "exit a subprocess task" do
    subprocess_task("subprocess task", process: "subprocess process")
  end

  defprocess "subprocess process" do
    user_task("user task", group: "admin")
  end

  defevent "exit subprocess task",
    process: "exit a subprocess task",
    exit_task: "subprocess task",
    selector: &ME.exit_subprocess_task_event_selector/1 do
      prototype_task("prototype task 1")
      prototype_task("prototype task 2")
  end

end

Now start an iex session on your project and paste in the following:
alias Mozart.ProcessEngine, as: PE
alias Mozart.ProcessService, as: PS
alias Phoenix.PubSub
MyBpmApplication.load()
{:ok, ppid, uid, business_key} = PE.start_process("exit a subprocess task", %{})
PE.execute(ppid)

And you should see the following:
iex [10:21 :: 1] > alias Mozart.ProcessEngine, as: PE
Mozart.ProcessEngine
iex [10:21 :: 2] > alias Mozart.ProcessService, as: PS
Mozart.ProcessService
iex [10:21 :: 3] > MyBpmApplication.load()
{:ok, <content deleted for clarity>}
iex [10:21 :: 4] > {:ok, ppid, uid, business_key} = PE.start_process("exit a subprocess task", %{})
10:21:26.446 [info] Start process instance [exit a subprocess task][847016f9-4818-4a30-9f65-3c9a5dcc1db5]
{:ok, #PID<0.296.0>, "847016f9-4818-4a30-9f65-3c9a5dcc1db5",
 "b085931c-89e5-48ca-ae81-e62ea486aef6"}
iex [10:21 :: 5] > PE.execute(ppid)
:ok
10:21:26.450 [info] New subprocess task instance [subprocess task][01dd052c-6cf2-4b95-b2b8-69418f3aa332]
10:21:26.450 [info] Start process instance [subprocess process][df6eca80-8c08-48ff-86cc-d825e8d7375f]
10:21:26.450 [info] New user task instance [user task][827cb198-b40b-4e78-8c80-39a5cbe56a6e]
From the logs, we see that the subprocess started and a user task was opened.
Now we will send an event that will cause the subprocess to exit. Additionally, the top level subprocess task will be completed and the tasks on the event will be executed:
PubSub.broadcast(:pubsub, "pe_topic", {:event, :exit_subprocess_task})
    
You should see:
iex [10:21 :: 7] > PubSub.broadcast(:pubsub, "pe_topic", {:event, :exit_subprocess_task})
:ok
10:28:18.310 [info] New prototype task instance [prototype task 1][985d41a3-3e5c-469a-847b-7295a6d405c0]
10:28:18.310 [info] Complete prototype task [prototype task 1]
10:28:18.310 [info] New prototype task instance [prototype task 2][a6394485-3a7f-464f-9a47-3463b6bf87be]
10:28:18.310 [info] Complete prototype task [prototype task 2]
10:28:18.310 [info] Process complete [exit a subprocess task][847016f9-4818-4a30-9f65-3c9a5dcc1db5]
Which indicates the expected behavior.



  

    
Mozart.BpmProcess 
    



      
This module implements a BPM Domain Specific Language (DSL) framework. To use, insert
use Mozart.BpmProcess into your modules.
defmodule Mozart.MyBpmApplication do
  use Mozart.BpmProcess

  # Your process model definitions
end

Mozart.Data.ProcessState 
    



      
Defines a struct with fields specifying the state of a BPM process execution. The fields
provide the following information:
	uid: A unique identifier of a single process model execution.
	business_key: A unique identifier for a hierarchy of process model executions
corresponding to a hierarchial process model.
	parent_uid: If the process is a child process, this is the pid of the parent process
execution.
	process: The name of the process model being executed.
	start_time: The time when the process started executing.
	end_time: The time when the process finished executing.
	execution_duration: The duration of time over which the process executed.
	open_tasks: a map of open tasks. The key is a process name. The value is the
internal structure of the task.
	completed_tasks: A list of completed tasks. List items are task internal
structures.
	data: A map of all accumulated process data. Keys are the property names. Values
are the property values.
	complete: Indicates whether the process has completed.

defstruct [
    :uid,
    :business_key,
    :parent_uid,
    :process,
    :start_time,
    :end_time,
    :execute_duration,
    open_tasks: %{},
    completed_tasks: [],
    data: %{},
    complete: false
  ]
end

Mozart.ProcessEngine 
    



      
A ProcessEngine is dynamically spawned for the purpose of executing a process model defined by defprocess function call.

Mozart.ProcessService 
    



      
This modeule provides a set of utility services.

